
Data



Primitive Data Types

Data Type Description Values
boolean truth values true, false

char text characters a-z,A-Z,0-9,etc.

byte very small whole numbers -128 to 127

short small whole numbers -32768 to 32767

int whole numbers -2147483648 to 2147483647

long big whole numbers -263 to 263-1

float real numbers 1.4e-45 to 3.4e+38

double high-precision real numbers 4.9e-324 to 1.8e+308



Variables

A variable is a container for storing data, every variable has two parts:

Name: identifies the variable

Value: value stored in the variable

Variables allow data to be reused

For example, if the number 21 is stored in a variable called age, then every time the word 
age is used in a program it will be represent (and be replaced with) the number 21.



Declaring Variables

In Processing, variables must be declared before they are used

Declaring a variable tells the computer the type of data that will be stored in the variable, 
and the variable name.

int x;     // Declare the variable x of type int 
float y;   // Declare the variable y of type float 
boolean b; // Declare the variable b of type boolean 
x = 50;    // Assign the value 50 to x 
y = 12.6;  // Assign the value 12.6 to y 
b = true;  // Assign the value true to b

As a shortcut, variables can be declared and assigned a value at the same time:
int x = 50;
float y = 12.6;
boolean b = true;



Naming Variables

In Processing, there are some restrictions on variable names: 

Variable names can only include a limited range of characters: a-z, A-Z, 0-9, _ 

Variable names cannot contain spaces, the underscore character “_” is often used instead 
of a space 

Variable names must not start with a number, although they can contain numbers in the 
middle and at the end

In Java (and hence Processing) programs, variable names typically begin with a 
lowercase letter and use a capital letter to start each new word 

e.g., width, rect4, fillColour, isFilled 

Variables that refer to constant data typically use all uppercase letters and 
underscores between each word 

e.g., PI, RED, MAX_VALUE



Choosing Good Variable Names

Variables should have names that describe their content.

This makes programs easier to read, reducing the need for lengthy comments and the 
potential for making mistakes

Imagine you need to create a variable to store room temperature, which of the 
following names would you choose?

t 

temp 

temperature 

roomTemp 

roomTemperature



Built-In Variables in Processing

Processing has built-in variables for storing important and commonly used data, e.g., 
the variables width and height store the width and height of the display window. 
Using width and height we can write sketches that scale to the size of the display 
window
size(300, 500); 
ellipse(width*0.5, height*0.5, width*0.66, height*0.66); 
line(width*0.5, 0, width*0.5, height); 
line(0, height*0.5, width, height*0.5);

Check the Processing reference for other interesting built-in variables, e.g., pmouseX, 
pmouseY, mouseX and mouseY provide the previous and current x and y position of 
the mouse in the sketch window. Try running the following code to see how easy it is 
to create a basic drawing program:
void draw() { line(pmouseX, pmouseY, mouseX, mouseY); }



Arithmetic



Arithmetic

Almost everything that the computer does is reduced at some point to numbers

All of the drawing functions take numbers to define the size and colour

We can use arithmetic to control the creation of sketches

The arithmetic doesn’t have to be difficult, it can be as simple as a little addition, 
subtraction and multiplication



Drawing with Arithmetic

int a = 30; 
line(a, 0, a, height); 
a = a + 40;
strokeWeight(4); 
line(a, 0, a, height);

int a = 8; 
int b = 10; 
line(a, 0, a, height); 
line(b, 0, b, height); 
strokeWeight(4); 
line(a*b, 0, a*b, height);



Drawing with Arithmetic

int y = 20; 
line(0, y, width, y); 
y = y + 6;
line(0, y, width, y); 
y = y + 6;
line(0, y, width, y); 
y = y + 6;
line(0, y, width, y);

float y = 20; 
line(0, y, width, y); 
y = y * 1.6;
line(0, y, width, y); 
y = y * 1.6;
line(0, y, width, y); 
y = y * 1.6; 
line(0, y, width, y);



Arithmetic Operators

Function Description Example
+ add 3 + 4
- subtract 6 - 5
* multiply 4 * 5
/ divide 10 / 2
% modulus 9 % 4



Data Types and Arithmetic Operators

Combining two integers will always result in an int, combining two floating-point 
numbers will always result in a float, but when an int and a float are combined 
the result is a float.

println(4/3);      // Prints "1" 
println(4.0/3);    // Prints "1.3333334" 
println(4/3.0);    // Prints "1.3333334" 
println(4.0/3.0);  // Prints "1.3333334"



Take Care Assigning Values

Integer values can be assigned to floating-point variables, but floating-point values 
cannot be assigned to integer variables

NOTE: The calculation is completed as an int before converting it to a float

int a = 4/3;        // Assign 1 to a 
int b = 3/4;        // Assign 0 to b 
int c = 4.0/3;      // ERROR! 
int d = 4.0/3.0;    // ERROR! 
float e = 4.0/3;    // Assign 1.3333334 to e 
float f = 4.0/3.0;  // Assign 1.3333334 to f



Operator Precedence

The order that arithmetic operators are applied is based on their precedence

Highest to lowest: % / * + - =

x = 3 + 4 * 5;    // Assign 23 to x 
y = (3 + 4) * 5;  // Assign 35 to y 



Increments and Decrements

Often we just want to add 1 to a value or subtract 1 from a value

We can do this easily with the increment (++) and decrement (--) operators

int x = 1; 
println(x);  // Prints "1" to the console 
x++;         // Equivalent to x = x + 1 
println(x);  // Prints "2" to the console
x--;         // Equivalent to x = x - 1 
println(x);  // Prints "1" to the console



Combination Operators

To add, subtract, multiply or divide a value stored in a variable and store the result in 
the same variable we can use the combined operator assignments

int x = 1; 
println(x);  // Prints "1" to the console 
x += 5;      // Equivalent to x = x + 5 
println(x);  // Prints "6" to the console
x -= 10;     // Equivalent to x = x - 10 
println(x);  // Prints "-4" to the console



Constraining Numbers

Processing provides a set of functions for constraining numbers

ceil()
floor()
round()
min()
max()

These functions are different from line() or ellipse() because they return values 
that can be assigned to a variable, or used in an arithmetic expression



Rounding Up and Down

The ceil() function rounds numbers up to the nearest whole number

w = ceil(2.0);  // Assign 2.0 to w 
x = ceil(2.1);  // Assign 3.0 to x 
y = ceil(2.5);  // Assign 3.0 to y 
z = ceil(2.9);  // Assign 3.0 to z

The floor() function rounds numbers down to the nearest whole number

w = floor(2.0);  // Assign 2.0 to w 
x = floor(2.1);  // Assign 2.0 to x 
y = floor(2.5);  // Assign 2.0 to y 
z = floor(2.9);  // Assign 2.0 to z

The round() function rounds number to the nearest whole number

w = round(2.0);  // Assign 2.0 to w 
x = round(2.1);  // Assign 2.0 to x 
y = round(2.5);  // Assign 3.0 to y 
z = round(2.9);  // Assign 3.0 to z



Minimum and Maximum

The min() function returns the smallest value of two or three numbers.

The max() function returns the largest value of two or three numbers.

int u = min(5,9);             // Assign 5 to u 
int v = min(-4, -12, -9);     // Assign -12 to v 
float w = min(12.3, 230.24);  // Assign 12.3 to w 
int x = max(5, 9);            // Assign 9 to x 
int y = max(-4, -12, -9);     // Assign -4 to y 
float z = max(12.3, 230.24);  // Assign 230.24 to z



Decisions



Relational Expressions

Relational expressions return Boolean values, i.e., the result of evaluating a relational 
expression is either true or false

Expression Evaluation
3 > 5 false
3 < 5 true
5 < 3 false
5 > 3 true



Relational Operators

Operators used in relational expressions compare two values to calculate a relation 
between them, e.g., is one greater than another

Operator Meaning
> greater than
< less than
>= greater than or equal to
<= less than or equal to
== equivalent to
!= not equivalent to



Conditionals

Conditionals allow a program to make decisions about which lines of code. They let 
actions take place only when a specific condition is met. 

Conditionals allow a program to behave differently depending on the values of variables. For 
example, the program may draw a line or an ellipse depending on the value of a variable.

The if structure is most commonly used in Processing to implement conditionals:

if (test) {
  statements
}

The test must be an expression that resolves to true or false.

When test evaluates to true, the statements inside the { (left brace) and } (right brace) 
are run. A sequence of statements between left and right braces is called a block and is a 
common feature in Processing.



Logical Operators

Boolean values can be combined or inverted by using logical operators

Operator Meaning
&& AND
|| OR
! NOT



Logical Expressions

Expression Evaluation
true && true true
true && false false
false && false false
true || true true
true || false true
false || false false

!true false
!false true



Repetition



line(10, 20, 10, 90);
line(15, 20, 15, 90);
line(20, 20, 20, 90);
line(25, 20, 25, 90); 
line(30, 20, 30, 90); 
line(35, 20, 35, 90); 
line(40, 20, 40, 90); 
line(45, 20, 45, 90); 
line(50, 20, 50, 90); 
line(55, 20, 55, 90); 
line(60, 20, 60, 90); 
line(65, 20, 65, 90); 
line(70, 20, 70, 90);

Iteration

Iteration can be used to greatly reduced the amount of code required to accomplish repetitive 
tasks.

int i = 10; 
while (i <= 70) {
  line(i, 10, i, 90);
  i += 5;
}



Iteration using while loops

The while structure is very much like the if structure, except that the statements 
inside the brackets are executed again and again until the test becomes false

while (test) {
  statements
}

In this example the statements inside the curly brackets will be executed as long as 
the value of x is less than 10. Notice that if the value of x did not increase inside the 
while loop, this loop would run forever.

int x = 0;
while (x < 10) {
  println(x);
  x++;
}



Using do...while Loops

A do...while loop is similar to a while loop, except that the test is performed after the 
code inside the brackets is run. Consequently, the code inside the brackets is always 
run at least once.

int x = 10;
do {
  println(x);
  x--;
} while (x > 1)



Iteration using for loops

Often we need to initialise, test and update a variable to complete a loop. The for 
structure allows these tasks to be put together and is structured like this:

for (init; test; update) {
  statements
}

The following example does the same as the example while loop we looked at earlier 
but using a for loop:

// An example for loop
for (int x = 0; x < 10; x++) {
  println(x);
}



EXERCISE

1. Create a sketch that uses a while loop to draw 20 rectangles in a row.

2. Create a sketch that uses a for loop to draw 10 circles in a vertical stack.

3. Translate one of the sketches in the “Drawing with Arithmetic” slide to use a loop, 
instead of having to repeat the same arithmetic and drawing statements multiple 
times.

3. Create a sketch that uses two for loops, one inside the other, to draw a matrix of 
10x10 rectangles.



Arrays



Data-Structures

Data-structure is a general term used to describe different ways of organising data in 
a computer’s memory so that they can be accessed or manipulated easily.

Some examples of common data-structures include arrays, lists, sets, maps and trees. 
We’re only going to look at the most common type: arrays



Arrays

An array is a way to store multiple items of data in memory so that they are accessed 
using a single variable

Each item of data in an array is given an index that can be used to refer to it.

Arrays can be used whenever a program needs to store multiple copies of same type of 
data in one place.

P R O C E S S I N G

0 1 2 3 4 5 6 7 8 9

Data

Index

An array of characters



Creating Arrays

Declaring a variable that points to an array is similar to declaring variables that point 
to primitive data types. The type of the data to be stored in the array is simply followed 
by square brackets.

int[] arrayOfInts;

Before an array can be used the memory needed to store the data has to be allocated. 
Memory is allocated using the “new” operator followed by the type of data to be stored and 
the number of items in the array:

arrayOfInts = new int[42];
arrayOfInts = new int[x + 5];
arrayOfInts = new int[width * height];

It is also possible to declare and allocate an array in a single statement:

int[] arrayOfInts = new int[42];

Unlike primitive data types however, this does not assign the values of the array. The 
program has to do this later by setting the value of each data item.



Assigning Values to Arrays

To access an item we use the variable name followed by the index in square brackets.

char[] vowels = new char[5];
vowels[0] = ‘a’;
vowels[1] = ‘e’;
vowels[2] = ‘i’;
vowels[3] = ‘o’;
vowels[4] = ‘u’;

Notice how the indices start at 0. This array with five items has indices 0...4.

It is possible to assign the values stored in an array don’t need to change using a 
special type of array allocation:

int[] arrayOfPrimes = {1, 2, 3, 5, 7};

This allocates an array big enough to hold five integers and initialises their values to the 
first five prime numbers.



Declaring, Creating and Assigning Arrays

int[] data;         // Declare
data = new int[5];  // Create
data[0] = 19;       // Assign
data[1] = 40;
data[2] = 75;
data[3] = 76;
data[4] = 90;

int[] data = new int[5];  // Declare and create
data[0] = 19;             // Assign
data[1] = 40;
data[2] = 75;
data[3] = 76;
data[4] = 90;

// Declare, create, and assign
int[] data = { 19, 40, 75, 76, 90 };



Array Index Out Of Bounds Exception

If you try to access a member of an array that lies outside the array boundaries, your 
program will terminate and give an ArrayIndexOutOfBoundsException.

int[] data = { 19, 40, 75, 76, 90 };
println(data[0]); // Prints 19 to the console
println(data[2]); // Prints 75 to the console
println(data[5]); // ERROR! The last element of the array is 4



Array Length

The length field stores the number of elements in an array.

int[] data1 = { 19, 40, 75, 76, 90 };
int[] data2 = { 19, 40 };
int[] data3 = new int[127];
println(data1.length); // Prints "5" to the console
println(data2.length); // Prints "2" to the console
println(data3.length); // Prints "127" to the console



Using Loops with Arrays

A for loop can be used to access array elements, especially with large arrays. The 
following example uses a for loop to iterate through every value in an array to draw 
lines.

int[] data = { 19, 40, 75, 76, 90 };
for (int i = 0; i < data.length; i++) {
  line(data[i], 0, data[i], 100);
}



Using Arrays to Draw

Arrays can make the task of programming much easier.

While it’s not necessary to use them, they can be valuable structures for managing data.

Arrays can be created to hold any type of data, and each element can be individually 
assigned and read.





Using Arrays to Draw

int[] x = {50, 61, 83, 69, 71, 50, 29, 31, 17, 39};
int[] y = {18, 37, 43, 60, 82, 73, 82, 60, 43, 37};

size(100, 100);
background(204);
strokeWeight(3);
beginShape();
for (int i = 0; i < x.length; i++) {
  vertex(x[i], y[i]);
}
endShape(CLOSE);



EXERCISE

Design a custom shape and encode the coordinates as arrays of values. Test your 
encoding using the following code, by inserting your values into the empty arrays:

int[] x = {};
int[] y = {};

size(100, 100);
background(204);
strokeWeight(3);
beginShape();
for (int i = 0; i < x.length; i++) {
  vertex(x[i], y[i]);
}
endShape(CLOSE);



Two-Dimensional Arrays

Data can also be stored and retrieved from arrays with more than one dimension. 

A 2D array is essentially an array of arrays. It must be declared, then created, and then the 
values can be assigned just as in a 1D array. The data points for the star can be put into a 
2D array instead of two 1D arrays:



Using Arrays to Draw

int[][] points = {
  {50, 18}, {61, 37},
  {83, 43}, {69, 60},
  {71, 82}, {50, 73},
  {29, 82}, {31, 60},
  {17, 43}, {39, 37}
};

size(100, 100);
background(204);
strokeWeight(3);
beginShape();
for (int i = 0; i < points.length; i++) {
  vertex(points[i][0], points[i][1]);
}
endShape(CLOSE);



EXERCISE

Translate your custom shape into a multidimensional array. Test your new encoding 
using the following code, by inserting your values into the empty array:

int[][] points = {
  // insert your point data here
};

size(100, 100);
background(204);
strokeWeight(3);
beginShape();
for (int i = 0; i < points.length; i++) {
  vertex(points[i][0], points[i][1]);
}
endShape(CLOSE);



Text



Characters

The char data type stores typographic symbols such as A, d, 5, and $.

The name char is short for character, chars are defined using single quotes.

char a = 'n';   // Assign 'n' to variable a 
char b = n;     // ERROR! Without quotes, n is a variable 
char c = "n";   // ERROR! The "" defines n as a String
char d = 'not'; // ERROR! Chars can hold only one character



Words and Sentences

Use the String data type to store words and sentences. Double quotation marks 
define a string.

The String data type is an object. String variables are declared similarly to int, float, 
etc., but the word String must be capitalised.

// The String data type can contain long and short text elements 
String s1 = "Rakete bee bee?"; 
String s2 = "Rrrrrrrrrummmmmpffff tillffff tooooo?"; 
println(s1); // Prints "Rakete bee bee?" 
println(s2); // Prints "Rrrrrrrrrummmmmpffff tillffff tooooo?"

// Strings can be combined with the + operator 
String s3 = "Rakete"; 
String s4 = "rinnzekete"; 
String s5 = s3 + s4; 
println(s5); // Prints "Rakete rinnzekete" 



The text() function is used to draw characters to the screen: 

text(data, x, y)
text(stringdata, x, y, width, height)

size(100, 100);
background(204);
fill(0);
text("LAX", 40, 30);
text("AMS", 40, 60);
text("FRA", 40, 90);

Drawing Text



Drawing Text with Word Wrapping

size(100, 100);
background(204);
fill(0); 
String s = "Response is the medium"; 
text(s, 10, 20, 80, 50);



Fonts



Fonts

Processing has a unique data type called PFont to store font data.

PFont is a class, which is a way that we can package up data and methods that relate to 
that data together in a single place. We’ll look at classes and objects later.



Fonts in Processing

To use a custom font in Processing (on the desktop), it can be first converted into the 
VLW format.

To manually convert a font, select the “Create Font” option from the Tools menu. Select a 
font from the list and click “OK.” The font generated and is copied into the current sketch’s 
data folder. To load the font we simply use the loadFont() function providing the name of 
the file we generated:

PFont font = loadFont(fontFilename);

To convert a font installed on the system from within a Processing sketch we use the 
createFont() function.

PFont font = createFont(fontName, size);
PFont font = createFont(fontFilename, size);

Determining whether to use loadFont() or createFont() may depend on the types of 
font that you want to use, but in general createFont() is more useful.



What Fonts Are Available?

When using createFont() we often need to find out 
what fonts are available on a system from within our 
code.

Processing provides a way to list all of the fonts installed 
on a system using the PFont.list() function:

size(400, 1000);
background(0);
fill(255);

String[] fontList = PFont.list();
println(fontList);
for (int i = 0; i < height/20; i++) {
  textFont(createFont(fontList[i], 18));
  text(fontList[i], 20, (i+1)*20);
}



EXERCISE

Try running the following code on your computer to see what fonts are installed:

size(400, 1000);
background(0);
fill(255);

String[] fontList = PFont.list();
println(fontList);
for (int i = 0; i < height/20; i++) {
  textFont(createFont(fontList[i], 18));
  text(fontList[i], 20, (i+1)*20);
}



Drawing Text with Fonts

To set the current font to use to draw text with we use the textFont() function.

size(100, 100);
background(204);
PFont font = createFont("Chunkfive.ttf", 32);
textFont(font);
fill(0);
text("LAX", 0, 40);
text("AMS", 0, 70);
text("FRA", 0, 100);



Drawing Text with a Font

size(100, 100);
background(204);
PFont font = createFont("Chunkfive.ttf", 16);
textFont(font);
size(100, 100);
background(204);
fill(0); 
String s = "Response is the medium"; 
text(s, 10, 20, 80, 50);



size(100, 100);
background(204);
PFont font = createFont("Chunkfive.ttf", 32);
textFont(font);
fill(255); // White 
text("DAY", 0, 40); 
fill(0); // Black 
text("CVG", 0, 70); 
fill(102); // Gray 
text("ATL", 0, 100);

Drawing Text with Colour



Drawing Text with Opacity

size(100, 100);
background(204);
PFont font = createFont("Chunkfive.ttf", 72);
textFont(font);
fill(0, 160); // Black with low opacity 
text("1",  0, 80); 
text("2", 15, 80); 
text("3", 30, 80); 
text("4", 45, 80); 
text("5", 60, 80);



Drawing Text with Multiple Fonts

size(100, 100);
background(204);
PFont font1 = createFont("rabiohead.ttf", 24);
PFont font2 = createFont("gooddog.otf", 24);
textFont(font1);
fill(0); 
textFont(font1); 
text("Rabiohead", 6, 45); 
textFont(font2); 
text("GoodDog", 2, 80);



Drawing Text at Different Sizes

size(100, 100);
background(204);
PFont font = createFont("Chunkfive.ttf", 32);
textFont(font);
fill(0); 
text("LNZ", 0, 40); // Large 
textSize(18); 
text("STN", 0, 75); // Medium 
textSize(12); 
text("BOS", 0, 100); // Small



EXERCISE

Create a sketch using one or more fonts to display the following text:

Isn't it funny
How a Bear likes honey?
Buzz! Buzz! Buzz!
I wonder why he does?

To find a suitable font, check these web sites:

http://www.fontsquirrel.com/

http://www.google.com/fonts/

Download a TTF or OTF version of your font and place it in your sketch’s “data” folder to 
make it available in Processing. The easiest way to do this is to drag’n’drop the font 
file on to your code.

http://www.fontsquirrel.com/
http://www.fontsquirrel.com/
http://www.google.com/fonts/
http://www.google.com/fonts/


Text Leading and Alignment

The textLeading() function sets the spacing between lines of text:

textLeading(dist) // dist is the distance between lines in pixels

The textAlign() function sets the text alignment:

textAlign(MODE) // MODE can be LEFT, CENTER, or RIGHT



Drawing Text with Leading

size(100, 100);
background(204);
PFont font = createFont("Chunkfive.ttf", 12);
textFont(font); 
String lines = "L1 L2 L3"; 
textLeading(10); 
fill(0); 
text(lines, 5, 15, 25, 100); 
textLeading(20); 
text(lines, 36, 15, 25, 100); 
textLeading(30); 
text(lines, 68, 15, 25, 100);



Drawing Text with Alignment

size(100, 100);
background(204);
PFont font = createFont("Chunkfive.ttf", 12);
textFont(font); 
line(50, 0, 50, 100); 
fill(0); 
textAlign(LEFT); 
text("LEFT", 50, 20); 
textAlign(RIGHT); 
text("RIGHT", 50, 40); 
textAlign(CENTER); 
text("CENTER", 50, 80);



Drawing Text with Alignment

size(100, 100);
background(204);
PFont font = createFont("Chunkfive.ttf", 12);
textFont(font); 
for (int y = 20; y <= 80; y += 20) { line(0, y, 100, y); }
fill(0); 
textAlign(CENTER, BOTTOM); 
text("BOTTOM", 50, 20); 
textAlign(CENTER, BASELINE); 
text("BASELINE", 50, 40); 
textAlign(CENTER, CENTER); 
text("CENTER", 50, 60); 
textAlign(CENTER, TOP); 
text("TOP", 50, 80);



Measuring Text

size(100, 100);
background(204);
PFont font = createFont("Chunkfive.ttf", 32);
textFont(font); 
fill(0); 
char c = 'U'; 
float cw = textWidth(c); 
text(c, 22, 40); 
rect(22,42, cw, 5); 
String s = "UC"; 
float sw = textWidth(s); 
text(s, 22, 76); 
rect(22,78, sw, 5);



EXERCISE

1. Use the text alignment functions to create a sketch that draws the number “8” in 
white at the center of a black circle.

2. Use the functions for measuring text to create a sketch that draws a rectangle 
around the word “rectangle” using the Pirulen font available for download from Font 
Squirrel at http://www.fontsquirrel.com/fonts/Pirulen

http://www.fontsquirrel.com/fonts/Pirulen
http://www.fontsquirrel.com/fonts/Pirulen


Questions? rob.saunders@sydney.edu.au

mailto:rob.saunders@sydney.edu.au
mailto:rob.saunders@sydney.edu.au

